

Datumsfunktionen!

Microsoft Excel speichert **Datumsangaben als ganze Zahlen**

und **Uhrzeitangaben als Nachkommazahlen**, da diese als Bruchteile eines Tages angesehen werden können (ein Tag hat daher den Zahlenwert 1, ein Stunde den 24sten Teil von 1 also $1/24 = 0,04166667$, 12 Stunden entsprechen 0,5 ...).

Datums- und Uhrzeitangaben sind Zahlen, also kann man damit rechnen.

Der erste Tag in Excels Zeitrechnung ist der 1.1.1900. Dieses Datum entspricht der Zahl 1.

Der 2.1.1900 ist die 2, der 3.1.1900 die 3, der 1.2.1900 ist die 32, der 29.2.1900* ist 60, 31.12.1900 ist 366, 26.9.1902 ist 1000, 6.7.2009 ist 40.000 ... (* den 29.2.1900 ... gab es nicht – Alle Daten danach berechnet Excel aber korrekt ...)

Man kann eine Datums- oder Uhrzeitangabe als fortlaufende Zahl anzeigen, indem man das **Zahlenformat** der Zelle, die das Datum oder die Uhrzeit enthält, in das **Format 'Standard'** ändert.

Shortcuts zum Einfügen des aktuellen Datums, so dass dieses Datum sich **nicht** bei jedem Öffnen des Dokuments aktualisiert:

Aktuelles Datum STRG + ÷ (Steuerung und Doppelpunkt)

Aktuelle Uhrzeit STRG + UMSCHALT + ÷

Formeln (Datumsangabe bitte ersetzen mit einem richtigen Datum oder mit einem Verweis auf eine Zelle mit richtigem Datum)

Aktuelles Datum: **=heute()** mit leerem Klammernpaar

Aktuelles Datum mit Uhrzeit: **=jetzt()** mit leerem Klammernpaar

Datum aus Einzelangaben zusammensetzen:

=Datum(Jahr-Wert;Monats-Wert;Tages-Wert)

z.B. =DATUM(2000;1;1) => 1.1.2000 wenn die Zelle *als Datum* formatiert ist, sonst 36526 *als Zahl*

oder: =Datum(2003;5;29) => Datum 29. Mai im Jahre 2003

oder: =Datum(2002;5+15;29+7) => Datum des 29. Mais, aber 15 Monate und 7 Tage später

Jahreszahl aus Datum auslesen: **=Jahr(Datumsangabe)**

z.B. =Jahr("28.1.1964") => 1964 (als Zahl oder Standard formatiert)

Bei der direkten Eingabe eines Datums muss dieses in Anführungszeichen eingeschlossen werden

oder: =Jahr(jetzt()) => aktuelles Jahr

oder: =Jahr(A3) sofern in A3 ein Datum steht, wird das Jahr daraus als Zahl zurückgegeben

Monatszähl aus Datum auslesen: **=Monat(Datumsangabe)**

z.B. =Monat("28.1.1964") => 1 (als Zahl oder Standard formatiert)

Bei der direkten Eingabe eines Datums muss dieses in Anführungszeichen eingeschlossen werden

Tageszahl aus Datum auslesen: **=Tag(Datumsangabe)**

z.B. =Tag("28.1.1964") => 28 (als Zahl oder Standard formatiert)

Bei der direkten Eingabe eines Datums muss dieses in Anführungszeichen eingeschlossen werden

Wochentag als Zahl zurückgeben: **=Wochentag(Datumsangabe)**

z.B. =WOCHENTAG("28.1.1964") => 3 (als Zahl oder Standard formatiert) 3 entspricht dem Dienstag

Die Excel-Woche beginnt mit dem Sonntag (=1) und endet mit dem Samstag(=7)

Datum aus Einzelangaben zusammensetzen Teil 2: **=Datum(Jahr-Wert;Monats-Wert;Tages-Wert)**

z.B. =Datum(Jahr(heute());12;24) => Diesjähriges Weihnachten

oder: =Datum(Jahr(heute());Monat(A1);Tag(A1)) => Diesjähriger Geburtstag

oder: =Wochentag(Datum(Jahr(heute()+10;12;31)) => Wochentag von Silvester in 10 Jahren

Bitte berechnen Sie:

- ▶ Wie viele Tage leben Sie schon?
- ▶ Wie viele Tage haben Sie gelebt, wenn Sie Ihren 66. Geburtstag feiern (und wie viele Schaltjahre haben Sie dabei erlebt)?
- ▶ In A1 soll die je aktuelle Uhrzeit der Öffnung der Mappe stehen.
- ▶ In A2 soll das Datum von heute stehen, das sich bei einem erneuten öffnen der Datei nicht verändert
- ▶ Wie viele Tage sind es noch bis Weihnachten?
- ▶ Auf welchen Wochentag fällt dieses Jahr Ihr Geburtstag?
- ▶ Wie viele Tage ist Ihr letzter Geburtstag schon her, in wie vielen Tagen ist Ihr nächster Geburtstag?



Wie viele Tage leben Sie schon?

Weil Excel jedes **Datum als Anzahl von Tagen** behandelt, können wir einfach den Geburtstag vom heutigen Datum abziehen und erhalten das gesuchte Ergebnis:

Tage vom 1.1.1900 bis zum Geburtstag	Gelebte Tage
	Tage vom 1.1.1900 bis heute

Mengenlehre: Der Menge aller Tage vom 1.1.1900 bis heute die Menge aller bis dahin nicht von Ihnen gelebten Tage (also Tage vom 1.1.1900 bis zu Ihrer Geburt) abziehen: Rest: Ihre bisher verlebten Tage

Formel: wenn in A1 Ihr Geburtstag steht, bzw. wenn der 15.8.1985 Ihr Geburtstag ist
= HEUTE()-A1 = HEUTE()-"15.8.1985"

Achten Sie darauf, das Zellenformat auf Standard zu stellen!

Wie viele Tage haben Sie gelebt, wenn Sie Ihren 66. Geburtstag feiern; wie viele Schaltjahre haben Sie dabei erlebt?

Gleiches Prinzip wie eben, nur dass wir nun das Datum Ihres Geburtstages von dem Datum Ihres 66. Geburtstages abziehen müssen. Letzteres erhalten wir, indem wir einfach beim Datum Ihres Geburtstages die Jahreszahl um 66 erhöhen

Formel: wenn in A1 Ihr Geburtstag steht schreiben Sie in B1:

=DATUM(JAHR(A1)+66;MONAT(a1);TAG(a1)) - A1

Datum Ihres 66. Geburtstages	Geburtstag
------------------------------	------------

Schaltjahre dabei in C1: =B1-365*66

ein Schaltjahr besitzt einen Schalttag mehr als normale Jahre. In B1 steht, wie viele Tag in den 66 Jahren vergangen seit A1 sind. Davon ziehen Sie die Tage von 66 Normal-Jahren ab. Für jedes Schaltjahr ergibt sich ein Rest von 1. Also wird in C1 die Anzahl der Schaltjahre angezeigt.

In A1 soll die je aktuelle Uhrzeit der Öffnung der Mappe stehen.

Die Uhrzeit entspricht der Nachkommastelle der Datumszahl. Ich erhalte sie, indem ich das, was vor dem Komma steht auf Null setze, indem ich das Datum von heute() von Datum mit Uhrzeit von jetzt() abziehe.

Formel: Zellenformat ,Uhrzeit'
= jetzt()-heute()

In A2 soll das Datum von heute stehen, das sich bei einem erneuten öffnen der Datei nicht verändert.

Der Wert darf nicht berechnet werden, weil er sich sonst bei jedem Öffnen des Dokuments verändern würde.

Entweder lassen Sie sich das aktuelle Datum mit =heute() ausgeben, kopieren es als Datum und fügen es anschließend mit Bearbeiten/ Inhalte einfügen/ Werte wieder ein. Stellen Sie das Zellformat auf ,Datum'

Formel:

Gehen Sie nach A2, drücken Sie STRG + Umschalt + Doppelpunkt

Wie viele Tage sind es noch bis Weihnachten?

Wie bei den Lebenstagen: Datum von Weihnachten künstlich erzeugen, davon das heutige Datum abziehen.

Formel:

= DATUM(JAHR(HEUTE());12;24)-HEUTE()

Auf welchen Wochentag fällt dieses Jahr Ihr Geburtstag?

Mit Wochentag-Formel den Wochentag bestimmen (und dann mittels der SVerweis-Funktion den richtigen Tag der Zahl zuordnen ...)

Formel: in A1 steht Ihr Geburtstag
=Wochentag(A1)

=SVERWEIS(WOCHENTAG(A1);WoTa;2)

Wenn im WoTa benannten Bereich die zweispaltige Wochentags-Matrix steht: 1 Sonntag
2 Montag ...

Wie viele Tage ist Ihr letzter Geburtstag schon her, in wie vielen Tagen folgt Ihr nächster?

Zuerst müssen Sie prüfen, ob Sie in diesem Jahr bereits Geburtstag hatten. Ihr diesjähriger Geburtstag ergibt sich aus dem aktuellen Jahr und Monat und Tag von Ihrem Geburtstag. Wenn Ihr Geburtstag schon vorbei ist, dann ist die heutige Datumszahl von Excel größer als die Datumszahl Ihres diesjährigen Geburtstags, weil heute weiter vom 1.1.1900 entfernt ist. Wenn Sie in diesem Jahr bereits Geburtstag hatten, dann müssen Sie dieses Datum vom heutigen Datum abziehen und das Zellformat mit Standard festlegen. Andernfalls müssen Sie Ihren Geburtstag des Vorjahres jahr(heute())-1 vom heutigen Datum abziehen.

Formel: wenn in A1 Ihr Geburtstag steht und in B1 Ihr diesjähriger Geburtstag

Ihr diesjähriger Geburtstag in B1: =DATUM(JAHR(HEUTE());MONAT(A1);TAG(A1))

Ihr Geburtstag im Vorjahr in C1: =DATUM(JAHR(HEUTE())-1;MONAT(A1);TAG(A1))

Bedingung: HEUTE()>B1 Dann: HEUTE()-B1 Sonst: heute()-datum(JAHR(b1)-1;MONAT(b1);TAG(b1))

Also: =WENN(HEUTE()>B1;HEUTE()-B1;HEUTE()-DATUM(JAHR(HEUTE())-1;MONAT(A1);TAG(A1)))

Oder: = WENN(HEUTE()>B1;HEUTE()-B1;HEUTE()-C1)

Der nächste Geburtstag:

Ihr Geburtstag im nächsten Jahr in D1: =DATUM(JAHR(HEUTE()+1;MONAT(A1);TAG(A1))

Bedingung: HEUTE()>B1 ('der Geburtstag war schon in diesem Jahr) Dann: D1- HEUTE() Sonst: B1- HEUTE()

Also: in E1 = WENN(HEUTE()>B1;D1-HEUTE();B1-HEUTE())

